Komplexe Zahlen und Iterationen, Level 1 bis 4

Mathematik mit MuPAD 2.5, Prof. Dr. Dörte Haftendorn 13.06.03 Version vom 16.06.03

```
Inhalt....: Komplexe Zahlen und Iterationen
Kategorie.: Lernblatt
Mathematik: Komplexe Zahlen, Fraktale Geometrie
MuPAD.....: 2.5.0
Datum.....: 2003-06-13
Autoren...: Dörte Haftendorn <Haftendorn@uni-lueneburg.de>
Funktionen: Re, Im, rectform, plot, plot::Point,
```

LEVEL 1

Quadratische Gleichungen haben nicht immer reelle Lösungen.

- `solve(x^2-6*x+13=0,x)`

$$\{3-2\cdot i,\ 3+2\cdot i\}$$

Hier haben sich zwei **komplexe Lösungen** ergeben. Komplexe Zaheln haben einen **Realteil**,
hier 3, und einen **Imaginärteil** als Faktor vor der **komplexen Einheit i.**

- `I^2   /* Eingabe des komplexen i  als großes I  */`

$$-1$$

- `Re(3-2*I), Im(3-2*I)   /*Realteil und Imaginärteil herausholen*/`

$$3,\ -2$$

-

Rechnen mit Komplexen Zahlen

- `3-2*I + 7-5*I    /* addiere Real- und Imaginärteil einzeln */`

$$10-7\cdot i$$

- `(3-2*I) * (7-5*I)   /* multipliziere mit den Klammerregeln */`

$11 - 29 \cdot i$

- `1/(3-2*I)     /* von Hand geht dies mit "Nenner-rational-machen" */`

$\frac{3}{13} + \frac{2 \cdot i}{13}$

Probe für die obige Gleichungslösung:

- `(3+2*I)^2-6*(3+2*I)+13`

0

-

Darstellen von komlexen Zahlen in der Gauß-Ebene

- `plot(plot::Point([0,0]),plot::Point([3,-2]),plot::Point([3,2]))`



Zwei Zahlen, die symmetrisch zur x-Achse liegen, heißen **konjugiert komplex**. Reelle Polynome haben außer evt. reellen Nullstellen nur solche konjugiert komplexen Nullstellenpaare.

- `z:=3-2*I;  w:=1+I/2`

$3 - 2 \cdot i$

$1 + \frac{i}{2}$

Oben sind die Koordinaten von Hand eingetragen, hier werden sie aus z und w bestimmt.

- `plot(plot::Point([0,0]),plot::Point([Re(z),Im(z)]),plot::Point([Re(w),Im(w)]))`



Noch komfortabler wird es, wenn man die folgende Prozedur benutzt:

```
• zpkt:=proc(z) begin
          return(plot::Point([Re(float(z)),Im(float(z))]))
      end_proc:
• plot(zpkt(0), zpkt(w), zpkt(1.5-I))
```



##

LEVEL 2   Darstellung in Polardarstellung

Der Abstand r einer komplexen Zahl z vom Ursprung heißt **Betrag von z abs(z)**.
Der Polarwinkel phi einer komplexen Zahl z heißt **Argument von z arg(z).**
Es gilt die Eulersche Formel

```
•  E^(i*`&phiv;`)  = cos(`&phiv;`)+I*sin(`&phiv;`)
```

$$e^{i \cdot \varphi} = \cos(\varphi) + i \cdot \sin(\varphi)$$

Polardarstellung

- ```
  z= r*E^(I*`ϕ`);
  ```

$$3 - 2 \cdot i = r \cdot e^{i \cdot \varphi}$$

- ```
  abs(z)*E^(I*arg(z))
  ```

$$\sqrt{13} \cdot e^{(-i) \cdot \arctan\left(\frac{2}{3}\right)}$$

-

Umwandlung aus der Polardarstellung in die rechtwinklige Darstellung:

- ```
  rectform(2*E^(I*PI/4))
  ```

$$\sqrt{2} + i \cdot \sqrt{2}$$

Polardarstellung aus Betrag und Winkel im Gradmaß:

- ```
  zbw:=proc(r,phi) begin
         return(r*E^(I*PI/180*phi))
       end_proc:
  ```
- ```
  zbw(2,35);rectform(%);float(%)
  ```

$$2 \cdot e^{\frac{7 \cdot i}{36} \cdot \pi}$$

$$2 \cdot \cos\left(\frac{7 \cdot \pi}{36}\right) + i \cdot \left(2 \cdot \sin\left(\frac{7 \cdot \pi}{36}\right)\right)$$

$$1.638304089 + 1.147152873 \cdot i$$

- ```
  plot(zpkt(0),zpkt(zbw(2,35)),zpkt(0.8*E^(I*0.2)))
  ```



•
##

Level 3 Iterationen auf Weg zum Apfelmännchen

Lineare Funktionen

```
f:=z->a*(z-fix)+fix;
a:=zbw(0.9,20);fix:=2+I; f(z)
```

$$z \to a \cdot (z - \mathrm{fix}) + \mathrm{fix}$$

$$0.9 \cdot \mathrm{e}^{\frac{\mathrm{i}}{9} \cdot \pi}$$

$$2 + \mathrm{i}$$

$$(0.9 - 2.7 \cdot \mathrm{i}) \cdot \mathrm{e}^{\frac{\mathrm{i}}{9} \cdot \pi} + 2 + \mathrm{i}$$

```
f(fix)   /* Probe 2+i ist wirklich Fixpunkt *)
```

$$2 + \mathrm{i}$$

```
ap:=zpkt(a): ap::Color:=[0,1,0]:
liste:=(f@@i)(w) $ i=1..40:/*Orbit von w*/
pkte:=map(liste,zpkt):
plot(zpkt(0),pkte,ap,zpkt(w),Scaling=Constrained)
```



•
Der Orbit für einen anziehenden Fixpunkt liegt auf einer Spirale mit dem Fixpunkt als Zentrum.
##
#####

Level 4 Quadratische Iterrationen mit f(z)=z^2 als Vorbereitung für Apfelmännchen

- ```
  f:=z->z^2:    f(z)
  ```

  $5 - 12 \cdot i$

- ```
  w:=-0.999-0.01*I;
  liste:=(f@@i)(w) $ i=1..16:/*Orbit von w*/
  pkte:=map(liste,zpkt);
  plot(zpkt(0),pkte,zpkt(w),Scaling=Constrained)
  ```



0 ist anziehender Fixpunkt für Folgen mit abs(z0)<1 und der Orbit liegt auf einer
Spirale,
die in den Ursprung führt.
Jedoch wird der Polarwinkel von zn für z(n+1) verdoppelt, darum kann man die
Spirale nur am
Anfang gut erkennen.
###

Level 5,6,7 in Datei julia-apfel.mnb